

# NAG Library Routine Document

## G07AAF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G07AAF computes a confidence interval for the argument  $p$  (the probability of a success) of a binomial distribution.

### 2 Specification

```
SUBROUTINE G07AAF (N, K, CLEVEL, PL, PU, IFAIL)
  INTEGER          N, K, IFAIL
  REAL (KIND=nag_wp) CLEVEL, PL, PU
```

### 3 Description

Given the number of trials,  $n$ , and the number of successes,  $k$ , this routine computes a  $100(1 - \alpha)\%$  confidence interval for  $p$ , the probability argument of a binomial distribution with probability function,

$$f(x) = \binom{n}{x} p^x (1-p)^{n-x}, \quad x = 0, 1, \dots, n,$$

where  $\alpha$  is in the interval  $(0, 1)$ .

Let the confidence interval be denoted by  $[p_l, p_u]$ .

The point estimate for  $p$  is  $\hat{p} = k/n$ .

The lower and upper confidence limits  $p_l$  and  $p_u$  are estimated by the solutions to the equations;

$$\sum_{x=k}^n \binom{n}{x} p_l^x (1-p_l)^{n-x} = \alpha/2,$$

$$\sum_{x=0}^k \binom{n}{x} p_u^x (1-p_u)^{n-x} = \alpha/2.$$

Three different methods are used depending on the number of trials,  $n$ , and the number of successes,  $k$ .

1. If  $\max(k, n - k) < 10^6$ .

The relationship between the beta and binomial distributions (see page 38 of Hastings and Peacock (1975)) is used to derive the equivalent equations,

$$p_l = \beta_{k, n-k+1, \alpha/2},$$

$$p_u = \beta_{k+1, n-k, 1-\alpha/2},$$

where  $\beta_{a,b,\delta}$  is the deviate associated with the lower tail probability,  $\delta$ , of the beta distribution with arguments  $a$  and  $b$ . These beta deviates are computed using G01FEF.

2. If  $\max(k, n - k) \geq 10^6$  and  $\min(k, n - k) \leq 1000$ .

The binomial variate with arguments  $n$  and  $p$  is approximated by a Poisson variate with mean  $np$ , see page 38 of Hastings and Peacock (1975).

The relationship between the Poisson and  $\chi^2$ -distributions (see page 112 of Hastings and Peacock (1975)) is used to derive the following equations;

$$p_l = \frac{1}{2n} \chi_{2k, \alpha/2}^2,$$

$$p_u = \frac{1}{2n} \chi_{2k+2, 1-\alpha/2}^2,$$

where  $\chi_{\delta, \nu}^2$  is the deviate associated with the lower tail probability,  $\delta$ , of the  $\chi^2$ -distribution with  $\nu$  degrees of freedom.

In turn the relationship between the  $\chi^2$ -distribution and the gamma distribution (see page 70 of Hastings and Peacock (1975)) yields the following equivalent equations;

$$p_l = \frac{1}{2n} \gamma_{k, 2; \alpha/2},$$

$$p_u = \frac{1}{2n} \gamma_{k+1, 2; 1-\alpha/2},$$

where  $\gamma_{\alpha, \beta; \delta}$  is the deviate associated with the lower tail probability,  $\delta$ , of the gamma distribution with shape argument  $\alpha$  and scale argument  $\beta$ . These deviates are computed using G01FFF.

3. If  $\max(k, n - k) > 10^6$  and  $\min(k, n - k) > 1000$ .

The binomial variate with arguments  $n$  and  $p$  is approximated by a Normal variate with mean  $np$  and variance  $np(1 - p)$ , see page 38 of Hastings and Peacock (1975).

The approximate lower and upper confidence limits  $p_l$  and  $p_u$  are the solutions to the equations;

$$\frac{k - np_l}{\sqrt{np_l(1 - p_l)}} = z_{1-\alpha/2},$$

$$\frac{k - np_u}{\sqrt{np_u(1 - p_u)}} = z_{\alpha/2},$$

where  $z_\delta$  is the deviate associated with the lower tail probability,  $\delta$ , of the standard Normal distribution. These equations are solved using a quadratic equation solver (C02AJF).

## 4 References

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

Snedecor G W and Cochran W G (1967) *Statistical Methods* Iowa State University Press

## 5 Arguments

- 1: N – INTEGER *Input*  
*On entry:*  $n$ , the number of trials.  
*Constraint:*  $N \geq 1$ .
- 2: K – INTEGER *Input*  
*On entry:*  $k$ , the number of successes.  
*Constraint:*  $0 \leq K \leq N$ .
- 3: CLEVEL – REAL (KIND=nag\_wp) *Input*  
*On entry:* the confidence level,  $(1 - \alpha)$ , for two-sided interval estimate. For example CLEVEL = 0.95 will give a 95% confidence interval.  
*Constraint:*  $0.0 < \text{CLEVEL} < 1.0$ .

- 4: PL – REAL (KIND=nag\_wp) Output  
*On exit:* the lower limit,  $p_l$ , of the confidence interval.
- 5: PU – REAL (KIND=nag\_wp) Output  
*On exit:* the upper limit,  $p_u$ , of the confidence interval.
- 6: IFAIL – INTEGER Input/Output  
*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N < 1$ ,  
 or  $K < 0$ ,  
 or  $N < K$ ,  
 or  $CLEVEL \leq 0.0$ ,  
 or  $CLEVEL \geq 1.0$ .

IFAIL = 2

When using the relationship with the gamma distribution to calculate one of the confidence limits, the series to calculate the gamma probabilities has failed to converge. Both PL and PU are set to zero. This is a very unlikely error exit and if it occurs please contact NAG.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

For most cases using the beta deviates the results should have a relative accuracy of  $\max(0.5E-12, 50.0 \times \epsilon)$  where  $\epsilon$  is the *machine precision* (see X02AJF). Thus on machines with sufficiently high precision the results should be accurate to 12 significant figures. Some accuracy may be lost when  $\alpha/2$  or  $1 - \alpha/2$  is very close to 0.0, which will occur if CLEVEL is very close to 1.0. This should not affect the usual confidence levels used.

The approximations used when  $n$  is large are accurate to at least 3 significant digits but usually to more.

## 8 Parallelism and Performance

G07AAF is not threaded in any implementation.

## 9 Further Comments

None.

## 10 Example

The following example program reads in the number of deaths recorded among male recipients of war pensions in a six year period following an initial questionnaire in 1956. We consider two classes, non-smokers and those who reported that they smoked pipes only. The total number of males in each class is also read in. The data is taken from page 216 of Snedecor and Cochran (1967). An estimate of the probability of a death in the six year period in each class is computed together with 95% confidence intervals for these estimates.

### 10.1 Program Text

```

Program g07aafe

!      G07AAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: g07aaf, nag_wp
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: clevel, phat, pl, pu
      Integer                     :: ifail, k, n
!      .. Intrinsic Procedures ..
      Intrinsic                   :: real
!      .. Executable Statements ..
      Write (nout,*) 'G07AAF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Display titles
      Write (nout,*) ' Probability      Confidence Interval '
      Write (nout,*)

d_lp: Do
      Read (nin,*,Iostat=ifail) n, k, clevel
      If (ifail/=0) Then
         Exit d_lp
      End If

!      Calculate probability
      phat = real(k,kind=nag_wp)/real(n,kind=nag_wp)

```

```

!      Compute probability
      ifail = 0
      Call g07aaf(n,k,clevel,pl,pu,ifail)

!      Display results
      Write (nout,99999) phat, '( ', pl, ' ', pu, ' )'
      End Do d_lp

99999 Format (1X,F10.4,6X,A,F7.4,A,F7.4,A)
      End Program g07aaf

```

## 10.2 Program Data

```

G07AAF Example Program Data
1067      117      0.95      : N, K, CLEVEL
402       54       0.95

```

## 10.3 Program Results

```

G07AAF Example Program Results

Probability      Confidence Interval

0.1097          ( 0.0915 , 0.1300 )
0.1343          ( 0.1025 , 0.1716 )

```

---